import std.algorithm;
import std.array;
import std.conv;
import std.math;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

T read(T)() { return readln.chomp.to!T; }
T[] reads(T)() { return readln.split.to!(T[]); }
alias readint = read!int;
alias readints = reads!int;

int countAB(string s) {
    return cast(int) s.zip(s[1..$]).count!(e => e[0] == 'A' && e[1] == 'B');
}

int calc(string[] a) {
    int ans = a.map!(countAB).sum;

    int xa = 0; // ...A : A で終わる
    int bx = 0; // B... : B で始まる
    int ba = 0; // B..A : B で始まり、A で終わる
    foreach (s; a) {
        auto front = s[0];
        auto last = s[$-1];
        if (front == 'B' && last == 'A') ba++;
        else if (front == 'B') bx++;
        else if (last == 'A') xa++;
    }

    // ba をつなぐ
    if (ba > 0) {
        ans += ba - 1;
        // 両サイドに xa, bx をつける
        if (xa > 0) {
            ans++;
            xa--;
        }
        if (bx > 0) {
            ans++;
            bx--;
        }
    }

    // xa, bx でペアを作る
    ans += min(xa, bx);
    return ans;
}

void main() {
    int n = readint;
    string[] a;
    for (int i = 0; i < n; i++) a ~= read!string;
    writeln(calc(a));
}

